
11.  xe "DYNAMISCHE VARIABELEN"Dynamische variabelenPRIVATE 

11.1  Inleiding
Tot nu toe hebben we op twee manieren variabelen gedeclareerd: als xe "Globale variabelen"globale va​ria​be​len van een programma of een unit en als xe "Lokale variabelen"lokale variabe​len binnen een procedure, functie of unit. Deze laatste zijn lokale variabelen, omdat ze uitsluitend gebruikt kun​nen worden in de procedure, functie of unit waar ze gedeclareerd staan.

De twee soorten variabelen worden op verschillende manieren in het geheugen opge​sla​gen. Globale variabelen en constanten worden opgeslagen in het xe "Gegevenssegment"gegevensseg​ment (xe "Datasegment"datasegment). Het gege​vens​​seg​ment is een deel van het geheugen waar gegevens op​ge​​slagen worden en kan maxi​maal 64 Kb groot zijn. Gedurende de hele uitvoering van het pro​gram​ma blijven ze daar staan. Ze kunnen wel een andere waarde krijgen, maar niet verwijderd worden.

Lokale variabelen worden opgeslagen in de xe "Stack"stack. De stack is een gedeelte van het ge​heu​gen waarvan de afmeting wordt vastgesteld door de programmeur. De program​meur is gebonden aan een maximum van 64 Kb. De vertaling van "stack" is "xe "Stapel"stapel".

Gegevens worden op de stapel gelegd en daar weer vanaf gehaald op het moment dat dit nodig is. Hierbij wordt het principe gevolgd dat wat het laatste op de stapel ge​legd is, er ook weer als eerste afgehaald wordt. Als naar procedure A gesprongen wordt, dan wor​den de gegevens van A op de stack gezet. Springen we vanuit A naar de proce​dure B, dan worden de gegevens van de proce​dure B bovenop de gegevens van A gelegd. Bij het verlaten van de procedure B worden de bijbehorende gege​vens van de stack ge​haald. We zijn dan weer terug bij procedure A. De gegevens van A lig​gen nu weer bovenop op de stapel. Verlaten we procedure A, dan worden ook de ge​ge​vens van A van de stack gehaald. Hierdoor zijn de waarden in lokale varia​be​len vluch​tig. Na het verlaten van de functie of procedure is alles weg. Het op de stack zet​ten van gegevens noemt men een "xe "Push"

xe "Push"push", het van de stack afhalen heet een xe "Pop""xe "Pop"pop".

Het opslaan van variabelen in het gegevenssegment of op de stack heeft nadelen. Het kan zijn dat we met veel grotere gegevensstructu​ren willen werken dan 64 Kb. Als deze ruimte niet groot ge​noeg is, loopt ons pro​gramma vast. Turbo Pascal biedt ons de mogelijkheid om geheugen te reserveren voor variabelen, met deze variabelen te werken, en als we ze niet meer nodig hebben het geheugen weer vrij te geven. Als we in het verloop van het programma deze variabelen weer nodig hebben, dan clai​men we de ruimte opnieuw. Dit soort variabelen worden xe "Dynamische variabelen"dyna​mische variabelen ge​noemd, in tegenstelling tot xe "Statische variabelen"

xe "Statische variabelen"statische variabelen in het gegevenssegment. Het wer​ken met dynamische variabelen vereist een wat andere aanpak dan het werken met stati​sche variabelen, maar is absoluut niet zo moeilijk als vaak gesuggereerd wordt.

11.2  Werken met dynamische variabelen
Zoals gezegd is het geheugen verdeeld in xe "Adressen"adressen. Zo'n adres bestaat uit twee delen: het xe "Segmentgedeel"segmentdeel en het xe "Offsetdeel"offsetdeel. Samen vormen deze delen een adres. Om in het ge​heugen een adres op te slaan, is voor elk deel 2 bytes nodig.

Turbo Pascal levert een gegevenstype dat bestemd is om adressen in op te slaan. Dit ge​gevens​type wordt "xe "Pointer"pointer" genoemd. Om een volledig adres op te slaan, zijn 4 bytes nodig. Het adres dat in een pointervariabele staat kan het adres zijn van een veel grotere gegevens​structuur elders in het geheugen.

De plaats waar Turbo Pascal dit soort grotere gegevensstructuren opslaat is de xe "Heap"heap. Dit hebben we al in het vorige hoofdstuk kunnen lezen. xe "Heap"De heap is de overblijvende geheugenruimte nadat de programmacode, het gegevenssegment, de stack en een even​​tue​le overlay-buffer een plek hebben gekregen. Hoe groter de heap is, des te minder pro​blemen zullen we hebben om gegevens op te bergen.

Het programma POINT_1 laat zien hoe een record van 93 bytes op de heap opgesla​gen kan wor​den en door middel van een pointervariabele van 4 bytes bereikt kan wor​den. De pointervariabele staat in het gegevens​seg​ment, omdat deze gedeclareerd is als globale variabele. Als zij in een procedure of functie gedeclareerd zou zijn, dan zou bij het verlaten van de functie of procedure ook de variabele verdwenen zijn. Als het record niet meer nodig is, wordt dat stukje xe "Geheugen vrijgeven"geheugen weer vrijgegeven:

PROGRAM POINT_1;

USES CRT;

TYPE

   PLedenkaart = ^TLedenkaart;

   TLedenkaart = Record

      Naam     : String[30];

      Adres    : String[30];

      Plaats   : String[20];

      Postcode : String[7];

      Lidnummer: Word;

   END;

VAR

   LEDENKAART: PLedenkaart;

BEGIN

   ClrScr;

   Writeln('Beschikbaar geheugen is: ',MemAvail);

   New(LEDENKAART);

   Writeln('Beschikbaar nadat geheugen geclaimd is: ',

            MemAvail);

   WITH Ledenkaart^ DO

   BEGIN

      Naam      := 'J. Jansen';

      Adres     := 'Dorpsstraat 35';

      Plaats    := 'Ons Dorp';

      Postcode  := '1111 AA';

      Lidnummer :=  100

   END;

   WITH Ledenkaart^ DO

   Writeln('In ledenkaart staat nu: ',Naam,' ',Adres,

           ' ', Plaats,' ',Postcode,' ',Lidnummer);

   Dispose(LEDENKAART);

   Writeln ('Beschikbaar als geheugen vrijgegeven is:',

             MemA​vail);

   LEDENKAART := NIL;

   Readln

END.

Regels:Toelichting:
   [1]3-11Definieer TLedenkaart als een dynamisch record.

   [2]12-13
Declareer een dynamische variabele LEDENKAART van het type PLeden​kaart.

   [3]16

Laat met behulp van MemAvail het beschikbare geheugen zien.

   [4]17-19
Declareer geheugenruimte op de heap en laat opnieuw zien hoeveel geheugen er nu beschikbaar is.

   [5]20-27Zet gegevens in het record.

   [6]28-30

Laat zien wat er op het adres staat waar LEDENKAART naar wijst.

   [7]31-33
Geef de geheugenruimte op de heap vrij en laat zien hoeveel geheu​gen er beschikbaar is.

   [8]34

Laat LEDENKAART naar niets wijzen.

Toelichting:
   [1]Het eerste wat opvalt bij de definitie van het record is dat er een regel bijgekomen is:

PLedenkaart = xe "^"^TLedenkaart
Met deze opdracht geven we aan dat het type PLedenkaart een pointer naar het type TLeden​​kaart is. Het dakje is het symbool voor een pointer. TLeden​kaart is een record met de velden Naam, Adres, enzovoort.

   [2]De gedeclareerde variabele LEDENKAART heeft als typering PLedenkaart. De varia​bele LEDENKAART bevat dus niet zelf de gegevens. Het bevat slechts het adres op de heap waar de velden van het record –dat van het type TLedenkaart is– te vinden zijn.

   [3]Turbo Pascal beschikt over een functie xe "MemAvail"

xe "Memavail"MemAvail. Een aanroep van deze functie re​tour​​​​neert het aantal bytes beschikbaar geheugen.

   [4]We gebruiken nu de Turbo Pascal-procedure xe "New"

xe "New"New om ruimte op de heap te decla​reren. LEDEN​KAART is een pointer naar TLedenkaart. Als we LEDEN​KAART als para​meter aan New meegeven, dan zoekt New uit hoeveel bytes er op de heap nodig zijn om de velden van het type TLedenkaart te kunnen bevatten. Deze ruimte wordt ge​re​serveerd en het adres van de eerste byte wordt in LEDENKAART gezet. LEDEN​KAART bevat nu het adres van een record van het type TLedenkaart. Mem​Avail laat zien dat de hoeveelheid beschikbaar geheugen inderdaad kleiner geworden is.

   [5]Hier worden de velden van het record gevuld. Het veld Naam van het record is niet bereikbaar als LEDENKAART.Naam. LEDENKAART is immers geen record, maar een pointer. Het is echter wel een pointer die getypeerd is naar het type TLedenkaart. Het veld Naam kunnen we nu bereiken met Ledenkaartxe "^"^.Naam. We gebruiken hier weer het dakje. Ledenkaart^.Naam betekent: het veld Naam van het type TLedenkaart waar LEDENKAART naar wijst.

   [6]Zowel bij het vullen als bij het uitlezen van de velden moeten we het dakje gebrui​ken. Ook kunnen we hier het xe "WITH"WITH-statement gebruiken. WITH Leden​kaart^ DO maakt de velden van het type TLedenkaart waar LEDENKAART naar wijst direct toe​gan​ke​​lijk.

   [7]De Turbo Pascal-procedure xe "Dispose"

xe "Dispose"Dispose geeft de geheugenruimte weer vrij. Deze proce​du​re doet het tegenovergestelde van de procedure New. New reserveert geheu​gen​ruim​te en Dispose geeft de gereserveerde ruimte weer vrij. MemAvail rappor​teert hier dat het oorspronkelijke geheugen weer beschikbaar is.

   [8]Tenslotte geven we de variabele LEDENKAART de waarde xe "NIL"NIL. NIL is een pointer​adres dat ner​gens naar wijst. Wanneer Dispose het stukje geheugen heeft vrijgegeven, bevat LEDENKAART nog steeds het adres van de velden van TLedenkaart. Deze loca​tie is echter vrijgegeven. Het systeem zal de vrijgeko​men ruimte mogelijk op​nieuw wil​len gebruiken. Als je na Dispose een pointervari​abele uitleest, dan loop je grote kans dat je onbegrij​pelijke tekens op je scherm krijgt. Waar je wel voor moet op​​pas​sen, is dat je een pointerva​riabele de waarde NIL geeft nadat je Dispose aange​roept. Doe je dit niet, dan zal het adres van het gereserveerde geheugen verloren gaan. Je kunt dan niet meer bij de gegevens komen, omdat de computer niet meer weet waar ze in het geheugen staan. Omdat het adres verloren is gegaan, kan het ge​re​ser​veer​de geheugen ook niet meer vrijgegeven worden. 

NIL kan ook in een vergelij​king opgenomen worden, bijvoorbeeld in het geval je alleen iets naar een veld wilt schrijven als er ook inderdaad geheugen gereserveerd is. Bijvoorbeeld:

IF LEDENKAART <> NIL THEN ...
Er bestaat hiervoor zelfs een aparte Turbo Pascal-procedure:

IF xe "Assigned"Assigned(LEDENKAART) THEN ...
De procedure xe "Assigned"Assigned retourneert True als er voor LEDEN​​KAART geheugen gede​cla​reerd is.

Omdat het werken met dynamische variabelen misschien nogal wat vraagtekens op​roept, geef ik in het programma POINT_2 een tweede voorbeeld. Dit programma zet een array van 100 bytes op de heap. Na de declaratie van geheugenruimte krijgt elk element van de array een waarde. Tussentijds laat POINT_2 zien hoeveel geheu​gen er nog beschikbaar is en wat het groot​ste blok vrije ruimte is:

PROGRAM POINT_2;

USES CRT;

TYPE PGetallen = ^TGetallen;

     TGetallen = Array[1..100] of Byte;

VAR

   GETALLEN: PGetallen;

   I       : Byte;

BEGIN

   ClrScr;

   Writeln('Beschikbaar op heap:',MemAvail,

           ' Grootste vrije blok: ',MaxAvail);

   Writeln('Druk op Enter');

   Readln;

   GetMem(GETALLEN,SizeOf(TGetallen));

   Writeln('Beschikbaar na declaratie geheugen:',

            MemA​vail,' Grootste vrije blok: ',MaxAvail);

   Writeln('Druk op Enter');

   Readln;

   FillChar(Getallen^,SizeOf(TGetallen),0);

   FOR I := 1 TO 100 DO Getallen^[I] := I;

   FOR I := 1 TO 100 DO Writeln(Getallen^[I]);

   FreeMem(GETALLEN,SizeOf(TGetallen));

   GETALLEN := NIL;

   Writeln('Beschikbaar na vrijgeven geheugen:',

            MemA​vail,' Grootste vrije blok: ',MaxAvail);

   Readln

END.

Regels:     Toelichting:

   [1]3-4Definieer een dynamisch type PGetallen als een array van 100 bytes.

   [2]5-7Declareer de benodigde variabelen.

   [3]10-13
Toon het beschikbare geheugen en het grootste beschikbare blok.

   [4]14

Declareer het geheugen voor de dynamische variabele GETALLEN.

15-18Toon het beschikbare geheugen.

   [5]19
Zet alle elementen van de array op 0.

   [6]20

Vul de array Getallen^ met de waarden 1 tot en met 100.

21

Zet ieder element van Getallen^ op het scherm.

   [7]22

Geef de geheugenruimte van de variabele GETALLEN vrij.

   [8]23

Laat de variabele GETALLEN naar niets wijzen.

24-26Toon het beschikbare geheugen.

Toelichting:
   [1]Het dynamische type TGetallen is een Array [1..100] of Byte. Het spreekt vanzelf dat dit ook een array van records zou kunnen zijn, of welk gegevenstype dan ook. PGe​tal​len is gedefinieerd als pointer naar het type TGetallen. 

   [2]GETALLEN is een variabele van het type PGetallen. PGetallen is een pointer naar TGetallen.

   [3]In het programma POINT_1 hebben we MemAvail al eens gebruikt. MemAvail meldt het tota​le aantal bytes beschikbaar geheugen. Je kunt je ook voorstellen, dat door het de​cla​reren van geheugenruimte voor verschillende variabelen, en het vervolgens weer geheel of gedeeltelijk vrijgeven van die ruimte, het totale beschikbare geheugen zich niet voordoet als één aaneengesloten vrij blok. Tussen de verschillende geheugen​loca​ties die gebruikt worden voor gegegevensopslag, kan vrije geheugenruimte zitten. De Turbo Pascal-functie xe "MaxAvail"

xe "MaxAvail"MaxAvail geeft de afmeting van het grootste aaneengesloten blok vrije geheugenruimte terug.

   [4]In POINT_1 gebruikten we de Turbo Pascal-procedure xe "New"

xe "New"New voor het decla​re​ren van ge​heugen. De procedure New zoekt uit hoeveel bytes het doorgegeven gegevenstype be​slaat en roept vervolgens de Turbo Pascal-procedure xe "GetMem"GetMem aan. In POINT_2 wordt direct xe "GetMem"GetMem aangeroepen en wordt het aantal bytes doorgegeven als SizeOf​(TGe​tallen). De xe "SizeOf"

xe "SizeOf"SizeOf-functie geeft het aantal bytes terug dat noodzakelijk is om een array van 100 bytes op te slaan. We hadden hier ook SizeOf(Getallen^) kunnen gebruiken; hiermee wordt hetzelfde aantal bytes teruggeven. SizeOf(GETAL​LEN), dus zonder dakje, zou maar 4 bytes teruggeven. In dat geval vragen we namelijk om de afmeting in bytes van de pointervariabele GETALLEN. Een pointer gebruikt 4 bytes. Omdat New eerst moet uitzoeken hoeveel bytes er nodig zijn, is GetMem snel​ler.

   [5]Nadat het beschikbare geheugen getoond is, wordt met behulp van de Turbo Pascal- proce​dure xe "FillChar"

xe "FillChar"FillChar ieder element van de array op 0 gezet. Na de declaratie van een array bevat elk element van de array de gegevens die toevallig op die plaats in het ge​​heugen stonden. FillChar kan elk element op 0 zetten door bij de aanroep de te wij​zigen variabele en de afmeting van de te vullen variabele als VAR-parameter mee te geven. Daarachter wordt vermeld met welk letterteken de bytes gevuld moeten worden. Als het om een numeriek veld gaat, wordt de ASCII-waarde in het veld ge​zet. Op deze manier kunnen velden dus een beginwaarde van 0 tot en met 255 krij​gen.

Let erop dat de variabele als VAR-parameter Getallen^ doorgegeven wordt, en niet als GETALLEN. Als we GETALLEN door zouden geven, zou de pointer met nullen ge​​vuld worden. We zouden dan geen adres meer hebben en de computer zou vreem​​de dingen kunnen gaan doen.

   [6]In een FOR-lus die van 1 tot en met 100 loopt, wordt vervolgens de waarde van I in het correspon​derende element van Getallen^ gezet. Een element van de array wordt hier bereikt met Getallen^[I]. Ook hier is weer het dakje geplaatst. Het uitlezen van de elementen gebeurt op dezelfde wijze.

   [7]Nu we de array GETALLEN niet meer nodig hebben, kan het geheugen met behulp van Free​Mem vrijgegeven worden. xe "FreeMem"We hadden hier ook xe "Dispose"Dispose kunnen gebruiken, maar deze procedure zou eerst moeten uitzoeken hoeveel bytes er vrijgemaakt moeten worden, en zou vervolgens xe "FreeMem"FreeMem aanroepen. FreeMem moet naast de naam van de pointervariabele ook het aantal bytes door​krijgen dat vrijgegeven moet worden. 

   [8]Na het vrijgeven van het geheugen, geven we de pointervariabele GETALLEN de waar​de NIL om aan te geven dat zij op dat moment niet naar een geheugenlocatie wijst.

11.3  Het gaat wel eens fout
Vooral in het begin worden er fouten gemaakt, vooral omdat je geneigd bent te ver​geten dat dynamische variabelen niet met de gegevens zelf werken, maar met adres​sen die naar de gegevens wijzen. Ook het declareren en het weer vrijgeven van ge​heu​gen​ruimte leidt regelmatig tot fouten. In het volgende programma staat een fout die snel gemaakt wordt:


PROGRAM POINT_3;


USES CRT;


VAR


   NUMMER_1, NUMMER_2: ^Word;


BEGIN


   ClrScr;


   GetMem(NUMMER_1,SizeOf(Word));


   Nummer_1^ := 100;


   NUMMER_2 := NUMMER_1;


   Writeln('In Nummer_1^ staat:',Nummer_1^);


   Writeln('In Nummer_2^ staat:',Nummer_2^ );


   Writeln('Druk je nu op Enter dan gaat het fout.');


   Readln;


   FreeMem(NUMMER_1,SizeOf(Word));


   FreeMem(NUMMER_2,SizeOf(Word))


END.

In POINT_3 worden twee variabelen gedeclareerd als pointer naar het type Word. Voor de variabele NUMMER_1 wordt geheugen gedeclareerd. Op de plaats waar de varia​bele NUMMER_1 naar wijst, wordt het getal 100 gezet. Als we daarna opgeven:

NUMMER_2 := NUMMER_1;
dan wordt het adres dat in NUMMER_1 staat ook aan NUMMER_2 gegeven. In dat geval hebben we twee pointervariabelen die naar hetzelfde adres wijzen. Dat blijkt wel als we met een Writeln-opdracht de inhoud van Nummer_1^ en Nummer_2^ op het beeldscherm laten tonen. Ze geven beide de waarde 100. Als daarna de ruimte van NUMMER_1 vrijgegeven wordt, gaat nog steeds alles goed. Pas als we proberen om ook de geheugenruimte van NUMMER_2 vrij te geven, worden we gestraft met foutmelding 204: "invalid pointer operation". Natuurlijk heeft het systeem gelijk. We hebben maar één keer 2 bytes ruimte gedeclareerd. Dat was toen we NUMMER_1 naar xe "GetMem"GetMem stuurden. Daar​na hebben we het adres van NUMMER_1 in NUM​MER_2 gezet. Met andere woor​den: NUMMER_1 en NUMMER_2 wezen naar het​zelf​de adres op de heap. Als je dan vervolgens twee keer ruimte probeert vrij te geven die maar eenmaal gede​clareerd is, gaat het fout.

11.4  Opgaven
11.1
a. 
Declareer een dynamische variabele ZIN van het type String.

b. Declareer een dynamisch record met de velden Naam, Adres en Woon​plaats.


c. Declareer een dynamische Array [1..1000] of Char.


d. Declareer een dynamische Array [1..100] of Pointer.


e. Declareer een variabele Pointers als Array [1..100] of Pointer.

11.2Maak een programma dat dynamisch een Array [1..20] of Char decla​reert. Het geheugen voor deze array moet op de heap gereserveerd worden. De array moet volgezet worden met de letter 'A'. De inhoud van de array moet naar het scherm gestuurd worden. Daarna moet het geheu​gen vrij​ge​geven worden en moet de pointervariabele naar niets wijzen.

